

Un oscillateur à pont de Wien stabilisé par une CTN

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDF](#)

Un oscillateur à pont de Wien a besoin d'un procédé pour maintenir une forme sinusoïdale sur son signal de sortie. Sans ce procédé, soit il fournit un signal écrêté, soit il n'oscille pas. Parmi les différentes solutions, il existe l'emploi d'une résistance à coefficient de température négatif, appelé couramment CTN.

Ce document montre le modèle d'une CTN et, par des simulations Pspice, sa mise en œuvre dans l'oscillateur à pont de Wien.

1) L'oscillateur à pont de Wien

1.a Présentation, principe

Parmi les différentes façons de réaliser un oscillateur, il existe la solution d'utiliser un système bouclé, réglé volontairement à être instable. C'est le cas de l'oscillateur à pont de Wien, dont le fonctionnement est largement expliqué dans la littérature électronique. Nous rappelons ici le principe.


Dans l'oscillateur à pont de Wien, la chaîne directe est réalisée par :

- un montage à A. Op. dont l'amplification vaut $A = 1 + \frac{R_2}{R_1}$

- et d'une chaîne de retour formée d'un filtre de pont Wien dont la transmittance harmonique s'écrit $b(j\omega) = \frac{R C \omega}{3 R C \omega + j(R^2 C^2 \omega^2 - 1)}$

On remarque que l'amplificateur A est non inverseur.

Le montage oscillera à la fréquence pour laquelle $b(j\omega)$ présentera un déphasage nul si l'amplification donnée par A compense l'atténuation présentée par b à cette fréquence.



oscillateur à pont de Wien

Le tracé de la réponse harmonique du pont de Wien, ou un calcul, montre que le passage à 0° de la phase est à la fréquence $\frac{1}{2\pi R C}$ (quand la partie imaginaire est nulle). Dans ce cas, l'atténuation est $\frac{1}{3}$.

Le circuit oscillera si $A = 3$, soit $R_2 = 2 R_1$.

Néanmoins, ce régime oscillatoire n'est pas stable : soit l'amplitude de la sinusoïde diverge, soit converge vers 0, et l'oscillation s'arrête. C'est très facilement vérifiable en pratique, ou en simulation, par réglage de R_2 .

Application numérique

$R_1 = 3,3 \text{ k}\Omega$ $R_2 = 6,5 \text{ k}\Omega$ ou $6,8 \text{ k}\Omega$, soit une amplification de 2,97 ou 3,06.


$R = 15 \text{ k}\Omega$ $C = 6,8 \text{ nF}$, soit une fréquence d'oscillation attendue de 1560 Hz

A. Op. : LF411.

Rappelons qu'un signal sinusoïdal d'amplitude 15 V, de fréquence 1560 Hz, (soit 9800 rad/s), est délivrable par un A. Op. dont le slew rate est supérieur à $9800 \times 15 = 147\,000 \text{ V/s}$ soit $0,147 \text{ V}/\mu\text{s}$.

Le LF 411 est donné pour $13 \text{ V}/\mu\text{s}$ typique. On ne craint donc pas une triangularisation du signal.

1.b. Simulation (netlist en annexe : fichier wien.cir)




Oscillation si amplification supérieure à 3

Interprétation

Avec $R_2 = 6,5 \text{ k}\Omega$, l'amplification n'est que de 2,97, le système bouclé n'est pas instable. Tous les potentiels sont à 0 V.

Avec $R_2 = 6,8 \text{ k}\Omega$, l'amplification est de 3,06, le système bouclé part en oscillation. En pratique, le démarrage des oscillations est lié à des conditions initiales (mise en route de l'alimentation), à la valeur de l'amplification... Ici, en simulation, elles apparaissent franchement à partir de 80 ms. Sans conditions initiales précisées, c'est le bruit de calcul interne au simulateur qui génère un très léger signal qui finit par être amplifié. On remarque très nettement une divergence en loi exponentielle.



Zoom sur le démarrage des oscillations

Interprétation

L'amplitude des oscillations qui diverge est fatalement limitée par la tension maximale que peut délivrer l'A. Op. Ce dernier étant alimenté en +/- 15 V, nous avons les saturations à 14,17 V. L'onde n'est pas sinusoïdale. Par contre, en sortie du pont de Wien (broche e+ de l'A. Op.), l'onde s'approche d'un sinus, car les harmoniques présentes sont en partie éliminées par sa fonction passe bande.

Une mesure donne une période de 644 μs , soit une fréquence de 1,55 kHz.

En toute rigueur, la fréquence est très légèrement inférieure en signaux écrêtés que durant la phase pseudo sinusoïdale transitoire.

2) La CTN

2.a Présentation brève

Une « CTN » est une thermistance à coefficient de température négatif.

La loi entre la température et la résistance est donnée par la relation de Steinhart et Hart :

$$\frac{1}{T} = A + B \ln(R_T) + C [\ln(R_T)]^3 \quad R_T \text{ en Ohm, } T \text{ en Kelvin}$$

A, B, C sont des constantes caractéristiques du matériau. On peut trouver une relation faisant intervenir un terme en $[\ln(R_T)]^2$, mais son coefficient associé est très faible ce qui le rend négligeable. Cette relation est utilisée quand la CTN est exploitée en capteur de température : par une mesure de R_T , puis un calcul, on dispose de T.

La recherche de la fonction réciproque (R_T fonction de la température T) est très lourde de calculs.

Aussi, on adopte d'autres formulations, comme celle proposée par le constructeur Vishay :

$$R(T) = R_{ref} \exp\left(A + \frac{B}{T} + \frac{C}{T^2} + \frac{D}{T^3}\right) \quad \text{Autres A, B, C et T en Kelvin.}$$

R_{ref} est la valeur nominale à 298,15 K ($\equiv 25^\circ\text{C}$), qui s'écrit également R_{25} .

Par exemple, la CTN ci-contre, fabriquée par Vishay, a une valeur nominale de 6,8 k Ω .

Vishay, donne, dans sa data sheet,

$$\begin{aligned} A &= -14,6337 & B &= +4791,842 \\ C &= -115334 & D &= -3730535 \end{aligned}$$



Un paramètre caractérisant une CTN est l'indice de sensibilité thermique, noté β : $\beta = \frac{T_1 T_2}{T_2 - T_1} \ln\left(\frac{R_1}{R_2}\right)$

Où R_1, R_2 sont les valeurs de $R(T)$ à T_1, T_2 respectivement.

On peut donner alors une **relation simplifiée** dans cet intervalle de température :

$$R(T) = R_{ref} \exp\left(\beta \left(\frac{1}{T} - \frac{1}{T_{ref}}\right)\right)$$

T, T_{ref} en K

2.b modélisations

- Composant implanté en librairie

Dans l'installation de Pspice, il y a -entre autres- une librairie de CTN de Vishay, nommé themistor_vishay.lib.

Elle est organisée en un modèle générique (NTC_BASE), qui est un jeu d'équations dans un sous circuit, puis des modèles spécifiques propres pour attribuer les valeurs des paramètres à la CTN choisie.

Par exemple, en sélectionnant la NTCLE100E3682_B0, on retrouve les valeurs numériques de A, B, C, D, la valeur de R_{25} , ainsi que d'autres paramètres.

```
.SUBCKT NTCLE100E3682_B0 RN Rp PARAMS: TOLR=0 TOLB=0
X80 Rn Rp NTC_BASE Params:
+ w=-14.633719574957
+ x=4791.842
+ y=-115334
+ z=-3730535
+ gth=0.0068 gth1 = 0.0000833
+ cth=0.0476
+ a=-14.6337195749572
+ r25=6800
+ b=4791.842
+ c=-115334
+ d=-3730535
+ T0=273.15
+ TR={1+TOLR/100}
+ TB={1+TOLB/100}
.ENDS
```

Les paramètres Pspice de la NTCLE100E3682_B0

- **Sous circuit « CTN simplifiée »**

Reprenons l'indice de sensibilité thermique, β , entre 25 °C et 85 °C. Appliquons sa relation :


$$T_1 = 25 \text{ °C} = 298,15 \text{ K} \Rightarrow R_{25} = 6800 \exp \left(-14,63372 + \frac{4791,842}{298,15} - \frac{115334}{298,15^2} - \frac{3730535}{298,15^3} \right) = 6800 \text{ } \Omega$$

$$T_2 = 85 \text{ °C} = 358,15 \text{ K} \Rightarrow R_{85} = 6800 \exp \left(-14,63372 + \frac{4791,842}{358,15} - \frac{115334}{358,15^2} - \frac{3730535}{358,15^3} \right) = 727,811 \text{ } \Omega$$

$$D'o\grave{u} : \beta_{85/25} = \frac{298,15 \cdot 358,15}{60} \ln \left(\frac{6800}{727,811} \right) = 3977 \text{ K.}$$

Il est donc possible de créer un sous circuit, réalisant une résistance commandée par une tension, au travers la relation simplifiée : $6800 \exp \left(3977 \left(\frac{1}{E_{com} + 273,15} - \frac{1}{298,15} \right) \right)$, où

E_{com} est une tension image de la température, avec un coefficient unitaire : $1 \text{ V} \equiv 1 \text{ } ^\circ\text{C}$.




Si $E_{com} = 0 \text{ V}$, la CTN simplifiée est à la température 0 °C. Le calcul aboutit à 23,052 kΩ.

Si $E_{com} = 25,15 \text{ V}$, la CTN simplifiée est à la température 25,15 °C. La résistance est alors à 6,8 kΩ.

2.c. Simulation de test de la CTN (fichier test_CTN.cir)

Nous allons comparer les 2 modèles :


- Le modèle de la NTCLE100E3682_B0 de Vishay,
- Le modèle **CTN simplifiée**, représenté par un sous-circuit, à base d'une résistance commandée par une tension, elle-même image de la température.



Ces 2 composants sont placés dans un circuit afin de comparer leur caractéristique $R = f(T)$.

On fixe arbitrairement une tension d'alimentation de 10 V. $R_v = R_s = 6,8 \text{ k}\Omega$.

Nous effectuons une analyse DC, en fonction de la température, de 0 °C à 100 °C, par pas de 1 m°C.



Résistance fonction de la température dans [0 °C ; 100 °C]

Interprétation

Plaçons le curseur à :

- 0 °C : La CTN de Vishay présente 21,506 kΩ, la CTN simplifiée 23,05 kΩ, comme calculé,
- à 100 °C : La CTN de Vishay présente 459,7 Ω, la CTN simplifiée 465,7 Ω.

- sur la dynamique [25 °C ; 85 °C], les 2 courbes sont quasiment confondues.

On peut conclure que le modèle simplifié est fidèle au composant de Vishay.

- Plaçons le curseur sur 25 °C et 85 °C :
 CTN simplifiée, on retrouve le calcul : 6800 Ω et 727,8 Ω.
 CTN de Vishay, les valeurs sont très proches : 6641 Ω et 725,3 Ω.

X Values	25.000	85.000
V(1) / I(XCTNs:1)	6.8000K	727.833
V(10) / I(Xvishay:1)	6.6412K	725.338

- Sur ce run, on peut visualiser également les courants.
 Les 2 CTN sont parcourus par des courants sensiblement identiques, ce qui est cohérent avec 2 résistances identiques.

X Values	25.000	85.000
I(XCTNs:1)	735.294u	1.3258m
I(Xvishay:1)	743.982u	1.3289m

Courants pour Valim = 10 V

Faisons l'expérience de changer l'alimentation pour passer à 20 V.

- à 25 °C, le courant dans la CTN simplifiée est parcouru par 1,47 mA, alors qu'il passe 1,54 mA dans la CTN Vishay.

X Values	25.000	85.000
I(XCTNs:1)	1.4706m	2.6568m
I(Xvishay:1)	1.5396m	2.6603m

Courants pour Valim = 20 V

- à 85 °C, le courant dans la CTN simplifiée est parcouru par 2,66mA, alors qu'il passe 2,66 mA dans la CTN Vishay.

- La même information est donnée par les résistances respectives, la CTN Vishay présente à 6,19 kΩ (pour 25 °C avec I = 1,54 mA), et 718 Ω (pour 85 °C avec I = 2,66 mA), la CTN simplifiée est restée à 6800 Ω à 25 °C, et 727,8 Ω à 85 °C.

Cela montre que le modèle Pspice de la NTCLE100E3682_B0 tient compte également du courant pour déterminer la résistance.

2.d. Test statique CTN avec son propre échauffement (fichier test_CTN_echauffement.cir)

Bien que poussé, le modèle de Pspice ne tient pas compte de l'échauffement propre par effet Joule. C'est adapté avec l'emploi des CTN en capteur de température, usuellement parcourues par un courant très faible.


Or, dans notre application avec l'oscillateur, on cherche à détecter un niveau de tension grâce à l'élévation de température provoquée au sein de la CTN.

Pour simuler ce phénomène, il faut donc créer un modèle de CTN qui tient compte de l'élévation propre de la température, au travers la loi d'Ohm thermique.

Rappelons la loi d'ohm thermique :

R_{TH} est la résistance thermique (en °/W) entre le cœur de la CTN et l'air ambiant, P est le flux, la puissance « circulant » dans la résistance R_{TH}, qui provoque un écart de température ΔT. On a :


$$\Delta T = R_{TH} P \text{ avec } \Delta T \text{ en } ^\circ\text{C}, \quad R_{TH} \text{ en } ^\circ\text{/W}, \quad P \text{ en W.}$$



La température de la CTN est donc **calculée**, et non plus communiquée par la variable TEMP de PSpice. R_{TH} est donnée dans la data sheet par son inverse (facteur de dissipation thermique) δ = 7 mW/°C, d'où R_{TH} = 143 °C/W.

Remarque :

Sous Pspice, pour accéder, en cours de simulation, à un courant dans une branche, il est nécessaire de placer une source de tension nulle. Cette dernière a été ajoutée en série à R(T), nœud 2, afin de disposer de I, pour la source « H » (tension commandée par un courant) qui délivre V(3).



Le schéma de test est donc constitué d'un opérateur mathématique (source de tension commandée par une tension) qui calcule ΔT . Cet échauffement est ajouté à la température ambiante (25 °C).

On choisit de faire varier la tension d'alimentation de 0 à 15 V.
On fixe $R_S = 100 \Omega$.




Schéma de test pour la CTN soumis à son propre échauffement

Pour vérifier le comportement thermique, nous allons faire 3 tests :


- $R_{TH} = 0 \text{ °/W}$. Cela revient à simuler une CTN fixée de façon parfaite sur le plan thermique sur un dissipateur de dimension infinie. On simule donc la CTN à la température ambiante, quelle que soit la puissance qu'elle dégage. Sa résistance attendue est $R_{25} = 6,8 \text{ k}\Omega$.

- $R_{TH} = 143 \text{ °/W}$. C'est le composant laissé à l'air libre, à température ambiante de 25 °C, soudé sur le circuit.

- R_{TH} infiniment grande. Cela revient à la considérer comme parfaitement isolée de l'air ambiant, et donc sans aucune évacuation de la puissance : une faible puissance dissipée aurait comme conséquence une température tendant vers l'infini et donc une CTN dont la résistance tendrait vers $R_{ref} \exp(\beta(-\frac{1}{T_{ref}})) =$

$$6800 \exp(-3977/298) = 10,9 \text{ m}\Omega.$$

On fixe $R_{TH} = 1 \text{ MEG}$ arbitraire. On aura donc R_T légèrement supérieure à 10,9 mΩ.



Evolution de $R(T)$ selon la résistance thermique entre le coeur de la CTN et l'air ambiant

Interprétation :


Le comportement simulé est parfaitement cohérent :

- avec $R_{TH} = 0 \text{ °/W}$, R_{25} est constant = 6,8 kΩ.

- avec $R_{TH} = 143 \text{ °/W}$, R_{25} décroît avec l'échauffement propre de la CTN. A $Valim = 15 \text{ V}$, la résistance ne vaut plus que 5,2488 kΩ.

- avec $R_{TH} = 1 \text{ MEG °/W}$, on confirme que $R(T)$ décroît : on est à 77,3 mΩ en fin d'échelle.

Observons la puissance par $V(1) * V(3)$:



Evolution de la puissance selon la résistance thermique entre le coeur de la CTN et l'air ambiant (1 mV signifie 1 mW)

Interprétation


Pour le cas à $R_{TH} = 0$, on a $R(T) = 6,8 \text{ k}\Omega$ constant.

La puissance suit la loi de Joule : $P = U^2/R$, avec $U = V(1)$ et $R = 6800$. En fin de courbe, à $Valim = 15 \text{ V}$, le courant est de $2,1739 \text{ mA}$, $P = 32,136 \text{ mW}$.

Pour $R_{TH} = 143 \text{ °/W}$, $R(T)$ diminue comme montré en courbe précédente.

La loi de Joule devient $P = U^2/R(T)$, ce qui explique une puissance supérieure. A $Valim = 15 \text{ V}$, le courant est de $2,8 \text{ mA}$ et $P = 41,274 \text{ mW}$.

Observons l'élévation de température en $V(4)$:



Evolution de la température selon la résistance thermique entre le coeur de la CTN et l'air ambiant (1 V signifie 1 °C)

Interprétation

A $R_{TH} = 0$, on retrouve une élévation de température nulle. C'est un « court-circuit thermique ».

Pour $R_{TH} = 143 \text{ °/W}$, la dissipation de puissance s'accompagne d'une élévation de température. En fin de courbe, à $Valim = 15 \text{ V}$, $\Delta T = 5,9022 \text{ °C}$, soit une température de la CTN de $31,05 \text{ °C}$. On peut vérifier la loi d'Ohm thermique : $\Delta T = R_{TH} P$, soit $5,9022 = 143 \times 41,274$.

La courbe à $R_{TH} = 1 \text{ MEG}$ n'est pas représentée ici.

Pour information, en fin d'échelle, à $Valim = 15 \text{ V}$, le courant traversant la CTN est de $149,9 \text{ mA}$, sa tension est de $11,588 \text{ mV}$ car $R(T) = 77,3 \text{ m}\Omega$.

Dans ces conditions, la puissance n'est que de $1,7367 \text{ mW}$. Sa température est alors de 1737 °C . Cette élévation de température de 1712 °C avec moins de 2 mW montre bien une super isolation thermique...

2.e. Aspect dynamique de la CTN avec son propre échauffement

➤ Constante de temps thermique de la CTN


La constante de temps permet de connaître la réactivité de la CTN utilisée capteur de température. Elle caractérise un système du premier ordre, *linéaire*. Si le capteur est à dissipation nulle, lors d'un échelon de température extérieure ΔT , la température intérieure de la CTN suit la loi en $\Delta T [1 - \exp(-t/\tau)]$. Après un temps τ , constante de temps, la température interne a évolué de 63 %. On peut aussi indiquer qu'après 3τ , on n'est plus qu'à 5 % de la valeur finale.

Vishay indique 15 secondes de constante de temps thermique.

On peut raisonnablement penser, que, par réciprocity, si la CTN chauffe à une certaine température en interne, par loi de Joule, sa surface aura une loi en exponentielle de montée en température. Mais dans ce cas, la CTN n'est pas à dissipation nulle, et la constante de temps associée sera bien plus faible.

Une façon simple pour tenir compte de la constante de temps lors de la montée en température en interne, est de placer en amont de la commande de la CTN un circuit $R_f C_f$ passe bas :

Cela ne représente pas la constante de temps thermique, car on n'est pas dans les conditions de sa définition. Néanmoins, c'est un comportement équivalent.



Circuit $R_f C_f$ sur la commande de la CTN

Si on suppose qu'en 1 seconde, la CTN est à 95 % de sa température finale (ordre de grandeur estimé), on peut en déduire une constante de temps de 333 ms. En fixant $R_f = 1 \text{ k}\Omega$, on déduit $C_f = 333 \text{ }\mu\text{F}$.

➤ la CTN en signaux sinusoïdaux, avec son propre échauffement

Rappelons qu'en signaux variables, il y a lieu de considérer $i(t)$, $u(t)$ et la puissance $p(t) = u(t) i(t)$.

En signaux sinusoïdaux, de période T , de pulsation $\omega = 2\pi/T$:

Dans le cas général, on a : $u(t) = U_{\max} \sin \omega t$ et $i(t) = I_{\max} \sin(\omega t + \varphi)$, où $U_{\max} = U \sqrt{2}$ et $I_{\max} = I \sqrt{2}$.

En régime sinusoïdal entretenu (donc après un régime transitoire), la puissance efficace est donnée par :

$$P = \frac{1}{T} \int_0^T u(t) i(t) dt .$$

Après intégration, on retrouve la relation connue $P = U I \cos \varphi$.

Remarque : le signal temporel $p(t)$ est formé d'une composante continue, de valeur P , et d'une variation de pulsation 2ω .

Dans l'équation de la loi d'Ohm thermique, c'est P , la puissance efficace, qui est utilisée. Et ce, en statique. Dans notre schéma de simulation, il est alors nécessaire d'intercaler un montage intégrateur pour disposer de P , à partir de $u(t)$ et $i(t)$. Ce montage peut être un simple filtre passe bas qui prend la valeur moyenne et rejette la pulsation 2ω .

Notre oscillateur travaillant à 1,56 kHz, un simple circuit $R_f C_f$ passe bas tel que $\frac{1}{2\pi R_f C_f} \ll 3,12 \text{ kHz}$

peut convenir. En fixant $R_f = 1 \text{ k}\Omega$, on a $C_f \gg 0,051 \text{ }\mu\text{F}$. En prenant $10 \text{ }\mu\text{F}$, on respecte l'inégalité. La constante de temps est 10 ms. En sortie du filtre passe bas, en régime sinusoïdal, le signal sera représentatif de la puissance efficace P , mais après 30 ms environ.

Remarque : ce circuit $R_f C_f$ a deux rôles : modéliser une constante de temps thermique lors de montée en température par loi Joule, et modéliser l'action de l'intégration qui calcule P à partir de $p(t)$.

3) Oscillateur à pont de Wien équipé de la CTN

3.a schéma complet (fichier wien_asservi.cir)

C'est la fusion de l'oscillateur du pont de Wien, dans lequel on a retiré R_2 , et du montage à base de CTN simplifiée $R(T)$, qui remplace R_2 .

3.b. Choix de R_f et C_f


En simulation analogique, il est toujours délicat de faire cohabiter des signaux lents et des signaux rapides. En effet, les signaux rapides nécessitent un pas d'échantillonnage fin, et pour observer un potentiel qui évolue lentement, il faut une longue durée de simulation. La conséquence directe est un fichier de résultat volumineux, et dans les cas extrêmes, un « plantage » de Probe (logiciel de tracé de courbes) lors de l'accès au fichier.dat.

Remarque : il existe la possibilité de ne stocker qu'un nombre limité de potentiels en personnalisant la directive .PROBE. Cela permet de réduire la taille du fichier.dat

Aussi, de choisir $R_f = 1\text{ k}\Omega$ et $C_f = 10\text{ }\mu\text{F}$, soit une constante de temps de 10 ms, est cohérent pour déterminer P à partir de $p(t)$, mais donne un comportement plus rapide que celui donné par une constante de temps thermique de l'ordre de 333 ms.

La température ambiante est fixée à 25 °C.

3.c. Résultats de simulation




*Trace supérieure : installation du régime sinusoïdal en sortie de l'A. Op.
Trace inférieure : variation de résistance de la CTN correspondante*


Interprétation

Dès $t = 0$, la résistance de la CTN vaut 6,8 kΩ : on se retrouve dans les mêmes conditions qu'avec R_2 , résistance fixe. La sortie part en oscillation, enveloppée par une exponentielle divergente, visible à cette échelle vers 80 ms.

Comme précédemment, le signal est écrêté par les saturations. La CTN est donc alimentée par un signal de fort niveau, ce qui provoque une dissipation thermique, et donc une diminution de sa résistance. Dès lors, l'amplification diminue, et le signal perd en amplitude. Après un régime transitoire, environ 140 ms, l'oscillation est stabilisée, et ne présente aucune saturation :




Oscillateur à pont de Wien équipée de la CTN simplifiée



L'oscillateur fonctionne en régime sinusoïdal

On a une amplitude de 11,8 V et une fréquence de 1,56 kHz. L'onde est parfaitement sinusoïdale.

On peut apprécier le rôle du filtre Rf Cf :




Interprétation

Le signal V(4) est la sortie du multiplieur qui représente $p(t)$. A cette échelle des temps, on ne peut pas distinguer la fréquence 3,12 kHz. Après filtre passe bas, on récupère la valeur moyenne 675 mV, après un régime transitoire d'environ 140 ms.

En pratique, ce régime transitoire peut être différent. D'ailleurs, en augmentant la constante de temps Rf Cf, on modifie aussi la façon dont le signal de sortie arrive à son régime de sinus entretenu : le régime est oscillatoire faiblement amorti. En fait, le système équipé d'une CTN peut être modélisé par un schéma bloc montrant 2 boucles de retour :

- une contre réaction équipée d'un détecteur d'amplitude et d'un filtre passe bas agissant sur l'amplification,
- et la boucle de retour faite par le pont de Wien.



CONCLUSION

Ce document a présenté 2 aspects : le fonctionnement d'un oscillateur à pont de Wien, et la mise en œuvre d'une CTN. La particularité est d'avoir introduit la loi d'Ohm thermique pour tenir compte de l'échauffement propre de la CTN, ce que le modèle Pspice ne prévoit pas. On a pu ainsi simuler un fonctionnement cohérent de la stabilisation en amplitude.

Annexe : fichiers.cir

```
Oscillateur a pont de Wien
* fichier wien.cir
.lib eval.lib
* circuit :
Vplus 15 0 DC 15 ; pour les LF411
Vmoins 14 0 DC -15
*
.param R2=6.8k
R1 em 0 3.3k
R2 em S {R2}
Rw1 S 3 15k
Cw1 3 ep 6.8n
Rw2 ep 0 15k
Cw2 ep 0 6.8n
.STEP param R2 list 6.6k 6.8k
X1 ep em 15 14 S LF411

.TRAN 10u 100m 0m 10u

.PROBE
.END
```

Netliste, prête à simuler

```
resistance variable avec la temperature
* fichier test_CTN.cir

.LIB Themistor_vishay.lib

Valim alim 0 DC = 10

RV alim 10 6.8k
Xvishay 10 0 NTCLE100E3682_B0

.param RN=6.8k
.SUBCKT RCTNsimplifie Rout1 Rout2 in1 in2 ;
ETHERM Rout1 3 VALUE = { I(VSENSE)*{RN}*EXP( 3977/(V(in1,in2)+273.15) -
3977/(298.15) ) }
VSENSE 3 Rout2 DC 0
.ENDS

RS alim 1 6.8k
XCTNs 1 0 Com 0 RCTNsimplifie

ECom Com 0 VALUE = {TEMP}

.DC temp 0 100 1m

.probe
.end
```

Netliste, prête à simuler

```

Test CTN echauffement propre
* fichier test_CTN_echauffement.cir
* circuit :

Valim alim 0 DC=15
R2 alim 1 100
.temp=25
.param RN=6.8k
.SUBCKT RCTNsimplifie Rout1 Rout2 in1 in2; (273.15+{temp}+V(in3))
(273.15+{temp}))
ETHERM Rout1 3 VALUE = { I(VSENSE)*{RN}*EXP( 3977/(V(in1,in2)+(273.15+{temp})) -
3977/(298.15) ) }
VSENSE 3 Rout2 DC 0
.ENDS

.param RTH=143

XCTNsimplifie 1 2 4 0 RCTNsimplifie
Vbidon 2 0 DC=0 ; pour prélever le courant
HU 3 0 POLY(1) Vbidon 0 1 ; délivre U, image de I
EmultP 4 0 POLY(2) (1,0) (3,0) 0 0 0 0 {RTH}; délivre U I RTH

.DC Valim 0 15 1m
.STEP param RTH LIST 0 143 1MEG
.PROBE
.END

```

Netliste, prête à simuler

```

Oscillateur a pont de Wien
* fichier wien_asservi.cir
.lib eval.lib
* circuit :
Vplus 15 0 DC 15
Vmoins 14 0 DC -15
*
R1 1 0 3.3k
Rw1 S 6 15k
Cw1 6 7 6.8n
Rw2 7 0 15k
Cw2 7 0 6.8n
X1 7 1 15 14 S LF411

.temp=25
.param RN=6.8k
.SUBCKT RCTN Rout1 Rout2 in1 in2 ;
ETHERM Rout1 3 VALUE = { I(VSENSE)*{RN}*EXP( 3977/(V(in1,in2)+(273.15+{temp})) -
3977/(298.15) ) }
VSENSE 3 Rout2 DC 0
.ENDS

.param RTH=143
*
XCTN 1 2 5 0 RCTN
Vbidon 2 S DC=0 ; pour prélever le courant
HU 3 0 POLY(1) Vbidon 0 1 ; délivre U, image de I
EmultP 4 0 POLY(2) (1,S) (3,0) 0 0 0 0 {RTH}; délivre U I RTH
Rf 4 5 1k
Cf 5 0 10u
.TRAN 2u 200m 70m 2u
*
.PROBE
.END

```

Netliste, prête à simuler

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphasé sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)